Differentiation

Exercise 1

(a)

```
> q := sscanf("f*6#%\"xG6\"6#%*CopyrightGF%-%#ifG6%-%%typeG6$9$%)
  constantG-%&evalfG6#-%#LiG6#F.-.%\"qGF5F%F%F%","%m")[1];
```

$$q := \textbf{proc}(x)\ \ldots\ \textbf{end proc} \tag{1}$$

```
> plot(q(x),x=-1..10,-3..7);
```



The function is undefined for $x < 0$. It zero (and the graph is flat) at $x = 0$, then it drops down to $-\infty$ at $x = 1$, then climbs back to zero at about $x = 1.4$, and increases thereafter towards ∞.

(b)

When $x < 0$, the function $q'(x)$ is undefined (because $q(x)$ is). We have $q'(0) = 0$, and then $q'(x) < 0$ for $0 < x < 1$ (because the graph is sloping downwards). The derivative is undefined again when $x = 1$, but $q'(x) > 0$ for $x > 1$ (because the graph is sloping upwards).

(c)

```
> Q := (x,h) -> (q(x+h)-q(x))/h;
```

(2)

$$Q := (x, h) \mapsto \frac{q(x+h) - q(x)}{h} \quad \textbf{(2)}$$

```
> e := exp(1);
```

$$e := \mathrm{e} \quad \textbf{(3)}$$

```
> Digits := 30;
```

$$Digits := 30 \quad \textbf{(4)}$$

(d)

```
> Q(e^2,0.01);
```

$$0.499830983358648545443169703000 \quad \textbf{(5)}$$

```
> Q(e^2,0.0000001);
```

$$0.499999983083089748054000000000 \quad \textbf{(6)}$$

```
> Q(e^2,10^(-10));
```

$$0.499999999998308309000000000000 \quad \textbf{(7)}$$

It is clear from this that $q'(e^2) = \frac{1}{2}$.

(e)

```
> Q(e^3,10^(-10));
```

$$0.333333333333056738500000000000 \quad \textbf{(8)}$$

```
> Q(e^4,10^(-10));
```

$$0.249999999999942763000000000000 \quad \textbf{(9)}$$

It is clear from this that $q'(e^3) = \frac{1}{3}$ and $q'(e^4) = \frac{1}{4}$. We therefore guess that $q'(e^t) = \frac{1}{t}$ for all t. Moreover, given $x > 0$, we can write $x = e^t$ with $t = \ln(x)$, so $q'(x) = \frac{1}{\ln(x)}$. This means that $q(x)$ is actually the same as a standard function called $\mathrm{Li}(x)$, which is defined by $\mathrm{Li}(x) = \int_0^x \frac{1}{\ln(t)}\, \mathrm{d}t$; you can enter **?Li** for more information.

Exercise 2

(a)

The roots of $f'(x)$ occur where the graph of $f(x)$ is flat, or in other words, the tangent line is horizontal. This occurs wherever $f(x)$ has a local maximum or a local minimum (and possibly also in some other places, called inflexion points).

(b)

```
> f := (x)->x^3-x;
```

$$f := x \mapsto x^3 - x \quad \textbf{(10)}$$

```
> plot(f(x),x=-1.1..1.1);
```



There are roots of $f(x)$ at $x = -1$, $x = 0$ and $x = 1$. Between $x = -1$ and $x = 0$ we have a root of $f'(x)$ at about $x = -0.6$, corresponding to the top of the left-hand hump. Between $x = 0$ and $x = 1$ we have a root of $f'(x)$ at about $x = 0.6$, corresponding to the top of the right-hand hump. Thus, Rolle's principle is satisfied.

```
> solve(f(x)=0,x);
```

$$0, 1, -1 \tag{11}$$

```
> D(f)(x);
> solve(D(f)(x)=0,x);
```

$$3x^2 - 1$$

$$\frac{\sqrt{3}}{3}, -\frac{\sqrt{3}}{3} \tag{12}$$

```
> 3*x^2-1;
```

$$3x^2 - 1 \tag{13}$$

```
> plot([f(x),D(f)(x)],x=-1.1..1.1,-1..1);
```



(c)

```
> g := (x)->sin(x)+sin(3*x)/3;
```

$$g := x \mapsto \sin(x) + \frac{\sin(3\,x)}{3} \qquad \textbf{(14)}$$

```
> plot([g(x),D(g)(x)],x=-2*Pi..2*Pi);
```



We can see roots of $g(x)$ at $x = -2$ Pi, -Pi, 0, Pi and 2 Pi; in general, the roots are at $x = n$ Pi for integers n. Between $x = 0$ and $x = $ Pi there are two maxima and one local minimum, giving three roots of $g'(x)$. (This is perfectly consistent with Rolle's principle, which says only that there is *at least one* root of $g'(x)$ between 0 and Pi.) Similarly, between $x = $ Pi and $x = 2$ Pi there are two minima and one local maximum for $g(x)$, corresponding to the three places where the graph of $g'(x)$ (in green) crosses the x-axis. We can find the location of the these roots as follows:

```
> solve(D(g)(x)=0,x);
```

$$\frac{\pi}{4}, \frac{3\pi}{4}, \frac{\pi}{2} \qquad \textbf{(15)}$$

Maple has reported only the roots between 0 and Pi. We see from the graph that these roots can be shifted by any multiple of Pi, so the roots have the form $n + \frac{\text{Pi}}{4}$ or $n + \frac{\text{Pi}}{2}$ or $n + \frac{3\,\text{Pi}}{4}$ for integers n.

(We can get Maple to produce this answer by setting `_EnvAllSolution:=true;` however, it reports the result in a convoluted and confusing form, so it is better to just inspect the graph.)

(d)

Suppose that a and b are two different roots of $f(x)$, so $f(a) = 0$ and $f(b) = 0$. As x moves from a to b, the function $f(x)$ cannot increase all the time (otherwise $f(b)$ would be greater than 0) and it cannot decrease all the time (otherwise $f(b)$ would be less than 0). It must increase some of the time and decrease some of the time, so there must be some point at which it changes over from increasing to decreasing (or vice-versa), and at that point we will have $f'(x) = 0$.

(e)

As we see in the plot below, the roots of $h(x) = \tan(x)$ are at $x = n$ Pi for integers n.

```
> plot(tan(x),x=-Pi..Pi,-10..10,discont=true);
```



The graph is always sloping upwards, so $h'(x) > 0$ for all x. In fact, we have $h'(x) = \sec(x)^2 = 1 + \tan(x)^2$, which shows that $1 \leq h'(x)$ for all x. In any case, $h'(x)$ has no roots at all, contradicting Rolle's principle. So what goes wrong in the argument that we outlined? As x moves from 0 to Pi, the function increases from 0 up towards ∞, then jumps down discontinuously. When $x = \frac{\text{Pi}}{2}$, neither $h(x)$ nor $h'(x)$ is really well-defined.

```
> D(tan)(x);
```

$$\tan(x)^2 + 1 \qquad (16)$$

```
> plot(D(tan)(x),x=-Pi..Pi,0..10);
```



Exercise 3

(a)

Put $y = x^n$. Then $y' = n\,x^{n-1}$ and $y'' = n\,(n-1)\,x^{n-2}$ and $y''' = n\,(n-1)\,(n-2)\,x^{n-3}$. If $n = 0$ then $y' = 0$, so we cannot divide by y', so $S(y)$ is undefined. For any other value of n, we have $\frac{y''}{y'} = \frac{n-1}{x}$ and $\frac{y'''}{y'} = \frac{(n-1)\,(n-2)}{x^2}$. This gives

$$S(y) = \frac{y'''}{y'} - \frac{3\left(\frac{y''}{y'}\right)^2}{2} = \frac{(n-1)\,(n-2)}{x^2} - \frac{3\left(\frac{n-1}{x}\right)^2}{2} = \frac{2\,(n^2-3\,n+2) - 3\,(n-1)^2}{2\,x^2}$$
$$= \frac{1-n^2}{2\,x^2}$$

This is zero for $n = 1$ or $n = -1$.

(b)

```
> S := (y) -> diff(y,x,x,x)/diff(y,x) -
             (3/2)*(diff(y,x,x)/diff(y,x))^2;
```

$$S := y \rightarrow \frac{\frac{\partial^3}{\partial x^3} y}{\frac{\partial}{\partial x} y} - \frac{3}{2} \frac{\left(\frac{\partial^2}{\partial x^2} y\right)^2}{\left(\frac{\partial}{\partial x} y\right)^2} \qquad (17)$$

```
> S(x^n);
```

$$\frac{\left(\frac{x^n n^3}{x^3} - \frac{3 x^n n^2}{x^3} + \frac{2 x^n n}{x^3}\right) x}{x^n n} - \frac{3 \left(\frac{x^n n^2}{x^2} - \frac{x^n n}{x^2}\right)^2 x^2}{2 (x^n)^2 n^2} \qquad (18)$$

```
> simplify(S(x^n));
```

$$\frac{-n^2 + 1}{2 x^2} \qquad (19)$$

```
> solve(%=0,n);
```

$$-1, 1 \qquad (20)$$

(c)

```
> y := (a*x+b)/(c*x+d);
```

$$y := \frac{a x + b}{c x + d} \qquad (21)$$

```
> simplify(diff(y,x));
```

$$\frac{a d - b c}{(c x + d)^2} \qquad (22)$$

```
> simplify(diff(y,x,x));
```

$$-\frac{2 c (a d - b c)}{(c x + d)^3} \qquad (23)$$

```
> simplify(diff(y,x,x,x));
```

$$\frac{6 c^2 (a d - b c)}{(c x + d)^4} \qquad (24)$$

```
> simplify(S(y));
```

$$0 \qquad (25)$$

(d)

```
> z := (a*p^x+b)/(c*p^x+d);
```

$$z := \frac{a p^x + b}{c p^x + d} \qquad (26)$$

```
> simplify(S(z));
```

$$-\frac{\left(6 p^{2x} c^2 d^2 + p^{4x} c^4 + 4 p^{3x} c^3 d + 4 p^x c d^3 + d^4\right) \ln(p)^2}{2 \left(c p^x + d\right)^4} \qquad (27)$$

```
> 1/simplify(expand(1/%));
```

$$-\frac{\ln(p)^2}{2} \quad \textbf{(28)}$$

(e)

```
> T := (u) -> diff(u,x)^(1/2) * diff(diff(u,x)^(-1/2),x,x);
```

$$T := u \to \sqrt{\frac{\partial}{\partial x} u} \left(\frac{\partial^2}{\partial x^2} \left(\frac{1}{\sqrt{\frac{\partial}{\partial x} u}} \right) \right) \quad \textbf{(29)}$$

Now try some examples:

```
> w := x^n; simplify([S(w),T(w)]);
```

$$w := x^n$$

$$\left[\frac{-n^2+1}{2\,x^2}, \frac{n^2-1}{4\,x^2} \right] \quad \textbf{(30)}$$

```
> w := sin(x)^2; simplify([S(w),T(w)]);
```

$$w := \sin(x)^2$$

$$\left[\frac{-4\cos(x)^4+4\cos(x)^2-3}{2\sin(x)^2\cos(x)^2}, \frac{4\cos(x)^4-4\cos(x)^2+3}{4\sin(x)^2\cos(x)^2} \right] \quad \textbf{(31)}$$

```
> w := ln(x); simplify([S(w),T(w)]);
```

$$w := \ln(x)$$

$$\left[\frac{1}{2\,x^2}, -\frac{1}{4\,x^2} \right] \quad \textbf{(32)}$$

In each case we see that $T(w) = -\frac{S(w)}{2}$. In fact, this is true for any w. This can be shown as follows. We first use the chain rule to get:

$$\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x} u\right)^{-\frac{1}{2}} = -\frac{1}{2}\left(\frac{\partial}{\partial x} u\right)^{-\frac{3}{2}}\left(\frac{\partial^2}{\partial x^2} u\right)$$

We then differentiate once more, using the product rule and the chain rule, to get

$$\frac{\partial^2}{\partial x^2}\left(\frac{\partial}{\partial x} u\right)^{-\frac{1}{2}} = -\frac{1}{2}\left(-\frac{3}{2}\right)\left(\frac{\partial}{\partial x} u\right)^{-\frac{5}{2}}\left(\frac{\partial^2}{\partial x^2} u\right)^2 - \left(\frac{1}{2}\right)\left(\frac{\partial}{\partial x} u\right)^{-\frac{3}{2}}\left(\frac{\partial^3}{\partial x^3} u\right)$$

We now multiply both sides by $\sqrt{\frac{\partial}{\partial x} u}$ and simplify to get

$$\sqrt{\frac{\partial}{\partial x} u}\left(\frac{\partial^2}{\partial x^2}\left(\frac{\partial}{\partial x} u\right)^{-\frac{1}{2}}\right) = \left(\frac{3}{4}\right)\left(\frac{\partial}{\partial x} u\right)^{-2}\left(\frac{\partial^2}{\partial x^2} u\right)^2 - \left(\frac{1}{2}\right)\left(\frac{\partial}{\partial x} u\right)^{-1}\left(\frac{\partial^3}{\partial x^3} u\right)$$

or in different notation

$$T(u) = \frac{3\left(\frac{u''}{u'}\right)^2}{4} - \frac{u'''}{2\,u''} = \left(-\frac{1}{2}\right)\left(\frac{u'''}{u'} - \frac{3\left(\frac{u''}{u'}\right)^2}{2}\right) = -\frac{S(u)}{2}$$

as claimed.

Exercise 4

```
> y := cos(-2*ln(x));
```

$$y := \cos(2\ln(x)) \tag{33}$$

```
> plot(y,x=0..0.02);
```




```
> plot(y,x=0..1000);
```




```
> a[0] := exp(-Pi/4);
```

$$a_0 := e^{-\frac{\pi}{4}} \quad \textbf{(34)}$$

Here is the general formula for the derivative of $f(x)$:

```
> diff(y,x);
```

$$-\frac{2\sin(2\ln(x))}{x} \quad \textbf{(35)}$$

We can find the gradient at $x = a_0$ using the *subs*( ) command:

```
> m[0] := subs(x=a[0],diff(y,x));
```

$$m_0 := -\frac{2\sin\left(2\ln\left(e^{-\frac{\pi}{4}}\right)\right)}{e^{-\frac{\pi}{4}}} \quad \textbf{(36)}$$

However, it looks better if we simplify this:

```
> m[0] := simplify(subs(x=a[0],diff(y,x)));
```

$$m_0 := 2\,e^{\frac{\pi}{4}} \quad \textbf{(37)}$$

The equation of the line of slope m_0 passing through $[a_0, 0]$ is just $y = m_0\,(x - a_0)$.
We can plot this together with y as follows:

```
> plot([y,m[0]*(x-a[0])],x=0 .. 2*a[0]);
```



To find the other roots, note that $\cos(t) = 0$ precisely when t has the form $\left(k - \frac{1}{2}\right)$ Pi for some integer k.

Thus, we have $y = \cos(-2\ln(x)) = 0$ precisely when $-2\ln(x) = \left(k - \frac{1}{2}\right)$ Pi, so

$\ln(x) = \left(\frac{1}{4} - \frac{k}{2}\right)$ Pi,

so $x = e^{\left(\frac{1}{4} - \frac{k}{2}\right)\mathrm{Pi}}$. We write a_k for this root.

The steps that we did for a_0 can be repeated for a_3 as shown below. Then to do a_4 (for example) just change the 3 to 4 and press ENTER four times.

```
> k := 3;
```

$$k := 3 \qquad (38)$$

```
> a[k] := exp((1/4-k/2)*Pi);
```

$$a_3 := e^{-\frac{5\pi}{4}} \qquad (39)$$

```
> m[k] := simplify(subs(x=a[k],diff(y,x)));
```

$$m_3 := 2\,e^{\frac{5\pi}{4}} \qquad (40)$$

```
> plot([y,m[k]*(x-a[k])],x=0 .. 2*a[k]);
```



We find that the tangent lines always meet the y-axis at $y=2$ or $y=-2$. This is not too hard to see directly.

In general we have $\frac{dy}{dx} = -\frac{2\sin(2\ln(x))}{x}$. When $x=a_k$ we have $2\ln(x) = \left(\frac{1}{2} - k\right)$ Pi and so $\sin(2\ln(x)) = (-1)^k$, so $m_k = \frac{2(-1)^{k+1}}{a_k}$. The line L_k has equation $y = m_k(x - a_k)$ so the intercept at $x=0$ is $-m_k a_k$, which is $2(-1)^k$.